

Making Automatic Form Resizing Work for You



Colin Riddington
Mendip Data Systems



<https://isladogs.co.uk/automatic-form-resizing-1>

<https://isladogs.co.uk/using-afr>

Access DevCon – Fri 28 April 2023

Automatic Form Resizing (AFR) in Access

Questions:

Have you seen any of my previous presentations on AFR?

- Yes
- No

How often have you used AFR in your own apps?

- Regularly
- Occasionally
- Never

<https://forms.office.com/e/BCW2YfxTwS>

Automatic Form Resizing in Access

Why use automatic form resizing (AFR)?

- **Legacy apps** – text is now far too small to read at higher resolutions
- **Scale down** – forms are partly off screen on smaller monitors requiring scrolling
- **Alternatives** e.g. form layouts, grouping & anchoring give poor outcomes
- **User base** - monitors with a variety of screen sizes, form factors and resolutions
- **Client base** - have changed or will be changing their monitors
- **Future proofing** – ensuring that forms will still remain usable as circumstances change
- **Accessibility** - some users have eyesight issues and need to zoom their screens

When wouldn't AFR be useful?

- All users have exactly the same monitor size and resolution
- Situation is guaranteed to NEVER EVER change in the future

Automatic Form Resizing (AFR) in Access

Questions:


Have you seen any of my previous presentations on AFR?

- Yes
- No

How often have you used AFR in your own apps?


- Regularly
- Occasionally
- Never

Audience Responses to questions:



Automatic Form Resizing (AFR) in Access

Results



Automatic Form Resizing in Access

Before

Resize Form Example 2 (Click to Resize) Resize Form Now Form 1 Form 3 Form 4 Close

Screen Resolution / Shape: 1920 * 1080 : Widescreen (1.78) Refresh

*Click the **Resize Form Now** button in the form header*
All controls and their contents are repositioned and resized proportionately

Subform: **List of control types and available states (READ ONLY)**

ControlType	Show / Hide	Enable / Disable	Lock / Unlock
acBoundObjectFrame	Yes	Yes	Yes
acCheckBox	Yes	Yes	Yes
acComboBox	Yes	Yes	Yes
acCommandButton	Yes	Yes	Yes
acImage	Yes	No	No
acLabel	Yes	No	No
acLine	Yes	No	No
acListBox	Yes	Yes	Yes
acNavigationButton	Yes	Yes	No
acNavigationControl	Yes	Yes	No
acObjectFrame	Yes	Yes	Yes
acOptionButton	Yes	Yes	Yes
acOptionGroup	Yes	Yes	Yes
acPage	Yes	Yes	No
acPageBreak	No	No	No
acRectangle	Yes	No	No
acSubform	Yes	Yes	Yes
acTabCtl	Yes	Yes	No
acTextBox	Yes	Yes	Yes

Unbound Frame: **Control Property Count**

Automatic Form Resizing Example Version 3.93 03/03/2023 Colin Riddington, Mendip Data Systems 2005-2023

After

Resize Form Example 2 (Click to Resize) Resize Form Now Form 1 Form 3 Form 4 Close

Screen Resolution / Shape: 1920 * 1080 : Widescreen (1.78) Refresh

*Click the **Resize Form Now** button in the form header*
All controls and their contents are repositioned and resized proportionately

Subform: **List of control types and available states (READ ONLY)**

ControlType	Show / Hide	Enable / Disable	Lock / Unlock
acBoundObjectFrame	Yes	Yes	Yes
acCheckBox	Yes	Yes	Yes
acComboBox	Yes	Yes	Yes
acCommandButton	Yes	Yes	Yes
acImage	Yes	No	No
acLabel	Yes	No	No
acLine	Yes	No	No
acListBox	Yes	Yes	Yes
acNavigationButton	Yes	Yes	No
acNavigationControl	Yes	Yes	No
acObjectFrame	Yes	Yes	Yes
acOptionButton	Yes	Yes	Yes
acOptionGroup	Yes	Yes	Yes
acPage	Yes	Yes	No
acPageBreak	No	No	No
acRectangle	Yes	No	No
acSubform	Yes	Yes	Yes
acTabCtl	Yes	Yes	No
acTextBox	Yes	Yes	Yes

Unbound Frame: **Control Property Count**

Automatic Form Resizing Example Version 3.93 03/03/2023 Colin Riddington, Mendip Data Systems 2005-2023

Automatic Form Resizing in Access

What happens with AFR?


- Forms are automatically resized to match the user's screen size and resolution
- All form objects are repositioned proportionately as the form is resized
- All form objects and their contents are also resized proportionately
- Updates automatically if resolution or scaling changed and when moved to a different sized monitor
- Works with both overlapping windows and tabbed documents display modes
- Works with popup / split / navigation / datasheet forms and when application interface is hidden
- Optional form zoom and stretch / shrink features
- Works in both 32-bit & 64-bit Access
- Requires just ONE line of code in Form_Load event:

ResizeForm Me


Automatic Form Resizing in Access

What happens with AFR?

Design view




Form view



Automatic Form Resizing in Access

What happens with AFR?

Design view



Form view

Home Page for Adam TEACHER Mon 05/02/2018 Week 1 Tutor Group: 8QS

My Classes for Friday 02/02/2018 Click a lesson slot to view the class list. Double click to open the class attendance register View Today's Classes

8:00	9:00	10:00	11:00	12:00	13:00	14:00	15:00	16:00
AM Reg 8QS M32	Fri:2:1 11C/Es1 S09	Fri:2:2 10E/Sn2 S11			Fri:2:4 11EF/Tr1 S09	SOB E:07:00	Fri:2:5 11EH/Ds2 S09	

My Timetable

Week 1

1M	1Tu	1W	1Th	1F
1 Sci Nat 10E/Sn2 ACT S11	T Science 10EF/Tr2 ACT S09		Science 07BN/Sc ACT S09	Sci Nat 10E/Sn2 ACT T06
2 T Science 11EF/Tr1 ACT S09	Enterprise 11C/Es1 ACT S11 VIH 116595	Sci Dble 11EH/Ds2 ACT S09	Sci Dble 10CE/Ds1 ACT S09	
3 Sci Dble 10CE/Ds1 ACT S09	Science 09N/Sc2 ACT S09	Science 09N/Sc2 ACT S09	Sci Dble 10CE/Ds1 ACT S09	
4 Sci Dble 11EF/Ds1 ACT S09	Sci Nat 10E/Sn2 ACT S09	Sci Nat 10E/Sn2 ACT S09	Sci Dble 11EH/Ds2 ACT S09	
5 Sci Dble 11EF/Ds1 ACT S09	Science 09S/Sc4 ACT S09	Science 07BN/Sc ACT S09	Sci Dble 11EH/Ds2 ACT S09	

Coloured shading indicates TA support in the lesson

View week 2 Export Timetable to Outlook

School Calendar

Date	Day	Description	Start	End
05/02/2018	Mon	INSET DAY (4)		
06/02/2018	Tue	Literacy Month Yr 7	09:00	15:10
06/02/2018	Tue	School Photographs Year 11 only	09:20	15:10
07/02/2018	Wed	Literacy Month Yr 7	09:00	15:10
07/02/2018	Wed	Yr Mtgs	15:30	16:30
08/02/2018	Thu	MLDP 3	16:00	18:00
08/02/2018	Thu	Literacy Month Yr 7	09:00	15:10
09/02/2018	Fri	Literacy Month Yr 7	09:00	15:10
09/02/2018	Fri	End of Term 3		
19/02/2018	Mon	Start of Term 4		
19/02/2018	Mon	Literacy Month Yr 7	09:00	15:10
19/02/2018	Mon	LT Book Scrutiny Y10	09:00	16:00
19/02/2018	Mon	RTPA	15:30	16:30

Attendance Summary for 8QS

Pastoral Summary for 8QS

My 'Favourites' - Items most used

Item	Count
Enter Interim Grades	289
View Student Info	119
Log a new pastoral record	109
Parents Meeting Appointments	93
Pupil Support Plans	79
Class Lists	72
Class Photos	72
Tutor group lists	72
Tutor group photos	72
Send Student Group Email	44
View student grades	38
Multiple Commendation	37

Recent Referrals for 8QS

Name	Date	Code	From	To	Ack Done
BRISTOW, Samuel	05/02/2018	HW	OLP	TPB	<input type="checkbox"/>
CARVER, Daniel	05/02/2018	HW	OLP	TPB	<input type="checkbox"/>
DE BOER, Aniraj	05/02/2018	COM	ELF	ACT	<input type="checkbox"/>
ROSSITER, Rebecca	02/02/2018	COM	BMS	ACT	<input checked="" type="checkbox"/>
ROSSITER, Rebecca	02/02/2018	COM	LJJ	ACT	<input checked="" type="checkbox"/>
JACKSON, Jamie	01/02/2018	D	SJ	KT	<input checked="" type="checkbox"/>
SYLVESTER, Harry	31/01/2018	D	OLP	TPB	<input checked="" type="checkbox"/>

Click a class to view the class list or the * on the left to view class photos. Double click a class to open the interim report grade entry form.

Click to open any item in the list

Click the red box to view all 18 outstanding pastoral records marked for your attention

Click the small arrow at the top to view a full screen version of the pastoral summary with additional information.

Automatic Form Resizing in Access

Using zoom with AFR

75%

Form Zoom 75-125%

Save this zoom value for future use?

Resize Form Example Version 2.7 20/06/2021 Mendip Data Systems 2005-2021

100%

Form Zoom 75-125%

Save this zoom value for future use?

Resize Form Example Version 2.7 20/06/2021 Mendip Data Systems 2005-2021

125%

Form Zoom 75-125%

Save this zoom value for future use?

Resize Form Example Version 2.7 20/06/2021 Mendip Data Systems 2005-2021

Automatic Form Resizing in Access

Using AFR with datasheet form (DS)

Before resizing DS

Datasheet Resize Example Resize Datasheet Datasheet Zoom 100 Close

This demonstrates the use of AFR and zoom with a datasheet form (or subform).
 Datasheets are not affected by standard AFR code. The additional datasheet resizing code should normally be added at Form_Load rather than on a button click as used here for demonstration

ControlType	Show/Hide	Enable/Disable	Lock/Unlock	Resizable
acBoundObjectFrame	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acCheckBox	Yes	Yes	Yes	<input type="checkbox"/>
acComboBox	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acCommandButton	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acCustomControl	?	?	?	<input type="checkbox"/>
acImage	Yes	No	No	<input checked="" type="checkbox"/>
acLabel	Yes	No	No	<input checked="" type="checkbox"/>
acLine	Yes	No	No	<input type="checkbox"/>
acListBox	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acObjectFrame	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acOptionButton	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acOptionGroup	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acPage	Yes	Yes	No	<input checked="" type="checkbox"/>
acPageBreak	No	No	No	<input type="checkbox"/>
acRectangle	Yes	No	No	<input checked="" type="checkbox"/>
acSubform	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acTabCtl	Yes	Yes	No	<input checked="" type="checkbox"/>
acTextBox	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acToggleButton	Yes	Yes	Yes	<input type="checkbox"/>

Record: 1 of 19 No Filter Search

Automatic Form Resizing Example Version 3.4 05/11/2021 Colin Riddington, Mendip Data Systems 2005-2021

After resizing DS (+ 110% zoom)

Datasheet Resize Example UnResize Datasheet Datasheet Zoom 110 Close

This demonstrates the use of AFR and zoom with a datasheet form (or subform).
 Datasheets are not affected by standard AFR code. The additional datasheet resizing code should normally be added at Form_Load rather than on a button click as used here for demonstration

ControlType	Show/Hide	Enable/Disable	Lock/Unlock	Resizable
acBoundObjectFrame	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acCheckBox	Yes	Yes	Yes	<input type="checkbox"/>
acComboBox	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acCommandButton	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acCustomControl	?	?	?	<input type="checkbox"/>
acImage	Yes	No	No	<input checked="" type="checkbox"/>
acLabel	Yes	No	No	<input checked="" type="checkbox"/>
acLine	Yes	No	No	<input type="checkbox"/>
acListBox	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acObjectFrame	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acOptionButton	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acOptionGroup	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acPage	Yes	Yes	No	<input checked="" type="checkbox"/>
acPageBreak	No	No	No	<input type="checkbox"/>
acRectangle	Yes	No	No	<input checked="" type="checkbox"/>
acSubform	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acTabCtl	Yes	Yes	No	<input checked="" type="checkbox"/>
acTextBox	Yes	Yes	Yes	<input checked="" type="checkbox"/>
acToggleButton	Yes	Yes	Yes	<input checked="" type="checkbox"/>

Record: 1 of 19 No Filter Search


Automatic Form Resizing Example Version 3.4 05/11/2021 Colin Riddington, Mendip Data Systems 2005-2021

NOTE: Standard checkboxes CANNOT be resized

Automatic Form Resizing in Access


Using AFR with split form

Design view



NOTE: Remove splitter bar


Form view



Automatic Form Resizing in Access


Using AFR with navigation form

Design view



NOTE: Remove layout guides / anchoring


Form view




Automatic Form Resizing in Access

Using AFR with Access application interface hidden

Design view



Form view



Automatic Form Resizing in Access

Module modResizeForm declarations:

The screenshot shows the VBA editor for a project named 'Automatic Form Resizing Example'. The left-hand pane displays a tree view of the project's structure, including 'Microsoft Access Class Objects' and 'Modules'. The 'modResizeForm' module is selected and highlighted with a red box. The right-hand pane shows the 'Declarations' section of the module, which contains the following code:

```

'-----MODULE CONSTANTS & VARIABLES-----
'CHANGE THE VALUE BELOW TO THE HORIZONTAL 'BASE' RESOLUTION USED TO DESIGN YOUR FORM e.g. 800 X 600 -> 800
Private Const DESIGN_HORZRES As Long = 800 '3840 '1920 '1680 '1440 '1366 '1280 '1024 '800 '600

'CHANGE THE VALUE BELOW TO THE VERTICAL 'BASE' RESOLUTION USED TO DESIGN YOUR FORM e.g. 800 X 600 -> 600
Private Const DESIGN_VERTRES As Long = 600 '768 '1080 '1050 '900 '768 '600 '480
'NOTE: THE VALUE IS ONLY USED AS A CHECK in the GetCurrentFactor procedure

'CHANGE THE VALUE BELOW TO THE DPI SETTING USED TO DESIGN YOUR FORM
'Do not alter the DESIGN_PIXELS setting UNLESS you are sure
Private Const DESIGN_PIXELS As Long = 96
  
```

'Base' resolutions determine scaling factor

- Horizontal resolution = 1680 ; DESIGN_HORZRES = 800
- Scaling factor = $1680 / 800 = 2.1$

Automatic Form Resizing in Access

Typical AFR code:

```
Option Compare Database
Option Explicit

Private Sub Form_Load()
    ReSizeForm Me
End Sub

Private Sub Form_Open(Cancel As Integer)
    DoCmd.Maximize
End Sub
```

AFR subform code:

```
Option Compare Database
Option Explicit

Private Function IsSubform() As Boolean
    Dim bHasParent As Boolean

    On Error GoTo NotASubform

    ' If opened not as a subform, accessing
    ' the Parent property raises an error:
    bHasParent = Not (Me.Parent Is Nothing)

    IsSubform = True
    Exit Function

NotASubform:
    IsSubform = False
End Function

Private Sub Form_Close()
    UnReSizeForm Me
End Sub

Private Sub Form_Load()

    If Not IsSubform Then
        ReSizeForm Me
        Me.NavigationButtons = True
    End If
End Sub
```

Time to see AFR in action . . .



Automatic Form Resizing in Access

How to get the best results from AFR:

- Plan the use of AFR from the beginning – don't add to established application
- Always scale up . . . NOT down
- Develop in the lowest or 'base' resolution of any of your users
- Use a form template to match your base resolution (it should fill screen at that resolution)
- Make sure DESIGN_HORZRES value in modResizeForm matches base horizontal resolution
- Use maximised forms where possible
- Take care with popup forms
- Checkboxes, option buttons & button images cannot be resized (work-rounds for each)
- Don't use layout guides and anchoring
- Allow some blank space at bottom/right of form as a precaution – particularly if using zoom
- Keep tab controls and option groups near top / left of form to avoid 'over-size issues'
- Cannot exceed maximum allowed form width of 22.75 inches (57.79 cm).

Automatic Form Resizing in Access

Convert existing databases:

- Karl's Tips & Tricks: one 'over-large' form
- Better Date Picker: multiple re-draws
- Northwind 2000: many small forms of different types



Automatic Form Resizing in Access

Northwind 2000

- 21 forms of which 13 have code modules & none are maximised
- 2 resizable forms with datasheet sub forms (Customer Orders (x2) / Orders)
- 2 resizable forms with continuous sub forms (Categories / Quarterly Orders)
- 4 resizable single forms (Customers / Employees / Products / Suppliers)
- 1 non-resizable single form (Employees (page break))
- 2 resizable continuous forms (Customer Phone List / Product List)
- 1 resizable form with pivot table in unbound frame (Sales Analysis)
- 4 dialog forms (Customer Labels Dialog / Main Switchboard / Sales By Year Dialog / Sales Reports Dialog)
- 3 forms with embedded background images (Categories / Customers / Suppliers)
- 2 forms with photos in bound frames (Employees / Employees (page break))
- 1 modal popup form (Startup)



Automatic Form Resizing in Access

Resize code:

```

Public Sub ReSizeForm(frm As Access.Form)
    Dim rectWindow As tRect, sngFactor As Single
    Dim lngWidth As Long, lngHeight As Long

    On Error Resume Next

    CheckMonitorInfo 'check current monitor in use
    SetStatusBarText
    GetCurrentResolution 'get resolution info for current monitor|
    sngFactor = GetCurrentFactor 'get scaling factor for current monitor

    If lngH <> DESIGN_HORZRES Then 'resize necessary
        Resize sngFactor, frm 'local procedure

        'the following code controls the positioning of pop-up forms
        'but only if the form tag is null. This allows more control where it causes a problem
        If IsZoomed(frm.hWnd) = 0 Then 'Don't change window settings for maximised form.
            Access.DoCmd.RunCommand acCmdAppMaximize 'Max Access Window
            Call GetWindowRect(frm.hWnd, rectWindow)
            With rectWindow
                lngWidth = .Right - .Left
                lngHeight = .Bottom - .Top
            End With
            If Nz(frm.Tag, 1) <> 1 Then
                Call MoveWindow(frm.hWnd, ((lngH - _
                    (sngFactor * lngWidth)) / 2) - getLeftOffset, _
                    ((lngV - (sngFactor * lngHeight)) / 2) - getTopOffset, _
                    lngWidth * sngFactor, lngHeight * sngFactor, 1)
            End If
        End If
    End If

    'UseMDIMode property =1 (overlapping windows) or = 0 (tabbed documents)
    'next section fixes display issue for users of tabbed documents (MDIMode=0)
    If CurrentDb.Properties("UseMDIMode") = 0 Then
        MaximizeNavigationPane
        DoEvents
        MinimizeNavigationPane
    End If
End Sub

```

1

2

3

4

5

6

```

Private Sub Resize(sngFactor As Single, frm As Access.Form)
    Dim ctl As Access.Control, sngNCW As Single
    On Error Resume Next

    With frm 'Resize height for each section:
        .Width = .Width * sngFactor
        .Section(Access.acHeader).Height = .Section(Access.acHeader).Height * sngFactor
        .Section(Access.acDetail).Height = .Section(Access.acDetail).Height * sngFactor
        .Section(Access.acFooter).Height = .Section(Access.acFooter).Height * sngFactor
    End With

    'Resize and locate each control:
    For Each ctl In frm.Controls
        If ctl.ControlType <> acPage Then 'Ignore pages in TAB controls
            With ctl
                .Left = .Left * sngFactor
                .Top = .Top * sngFactor
                'CR 02/10/2021 - EITHER remove layout guide from navigation form OR reinstate height exclusion for acNavigationControl
                'If .ControlType <> acNavigationControl Then
                    .Height = .Height * sngFactor
                'End If
                If .ControlType = acNavigationControl Then 'CR 02/10/2021
                    .Width = .Width * sngFactor
                    sngNCW = .Width 'save width of navigation control
                ElseIf .ControlType = acNavigationButton Then 'do NOT use AFR scaling code - it would be MUCH too wide
                    .Width = sngNCW 'make width equal to navigation control
                Else
                    .Width = .Width * sngFactor
                End If

                'Modify certain combo box, list box and tab control properties:
                Select Case .ControlType
                    Case acLabel, acCommandButton, acTextBox, acToggleButton, acNavigationButton 'increase font size proportionately
                        .FontSize = .FontSize * sngFactor
                    Case acListBox 'as above & increase width of each column proportionately
                        .FontSize = .FontSize * sngFactor
                        .ColumnWidths = adjustColumnWidths(.ColumnWidths, sngFactor)
                    Case acComboBox 'as above & increase list width proportionately
                        .FontSize = .FontSize * sngFactor
                        .ColumnWidths = adjustColumnWidths(.ColumnWidths, sngFactor)
                        .ListWidth = .ListWidth * sngFactor
                    Case acTabCtl 'as for textbox & increase width/height of each tab proportionately
                        .FontSize = .FontSize * sngFactor
                        .TabFixedWidth = .TabFixedWidth * sngFactor
                        .TabFixedHeight = .TabFixedHeight * sngFactor
                    Case Else 'no code needed for other control types
                        'acRectangle, acCheckBox, acImage, acLine, acPageBreak, acSubform
                        'acOptionButton, acOptionGroup, acObjectFrame, acBoundObjectFrame, acNavigationControl
                End Select
            End With
        End If
    Next ctl
End Sub

```

Automatic Form Resizing in Access

Zoom form code:

1

```

Private Sub Form_Load()
  ReSizeForm Me

  blnZoom = True 'v3.79
  |
  'minimize nav pane
  MinimizeNavigationPane
  'minimize ribbon
  If IsRibbonMinimized = False Then ToggleRibbonState

  'set form footer captions
  Me.lblVersion.Caption = GetProgramName & " Version " & GetVersion & " " & GetVersionDate
  Me.lblMDS.Caption = GetCopyright

  'show screen resolution
  Me.lblResolution.Caption = GetResolution() & " : " & GetScreenShape() & " (" & GetFormFactor() & ")"

  'add new user record in tblUsers (if not already done) v2.7
  CurrentDb.Execute "INSERT INTO tblUsers ( UserName, DefaultZoom ) VALUES (GetUserName(), 100);"

  'initial zoom settings
  Me.SliderZoom.SetFocus
  Me.cboZoom = Nz(DLookup("DefaultZoom", "tblUsers", "UserName='" & GetUserName & "'"), 100)
  Me.SliderZoom.Value = Me.cboZoom

  If Me.cboZoom <> 100 Then
    Me.cmdResetZoom.Enabled = True
    Me.chkSaveZoom.Enabled = True 'v2.7
    sngOldZoom = 1 'v2.7 - needed to ensure zoom works correctly at form load
    UpdateFormZoom 'v2.7
  Else
    Me.cmdResetZoom.Enabled = False
    Me.chkSaveZoom.Enabled = False 'v2.7
  End If

  'save for future use
  sngOldZoom = Me.cboZoom / 100

End Sub

```

2

3

4

5

6

7

```

Private Sub UpdateFormZoom()
  Application.Echo False

  sngZoom = (Me.cboZoom / 100)
  If sngZoom <> sngOldZoom Then
    CentreForm Me
    ZoomForm Me
  End If

  'reset form section dimensions
  Me.Width = Me.LineFW.Width
  Me.FormHeader.Height = Me.LineHH.Height
  Me.Detail.Height = Me.LineDH.Height
  Me.FormFooter.Height = Me.LineFH.Height

  Select Case Me.cboZoom

  Case Is > 100
    Me.ScrollBars = 3
  Case 100
    Me.ScrollBars = 0
  Case Else
    Me.ScrollBars = 0
  End Select

  Application.Echo True

  'store zoom value for future reference
  sngOldZoom = Me.cboZoom / 100

End Sub

```

Automatic Form Resizing in Access

Stretch/Shrink form code:

```
Private Sub Form_Load()
    'get form dimensions on load
    GetOriginalSize Me

    'resize form depending on resolution & monitor size
    ResizeForm Me
End Sub
```

```
Private Sub Form_Resize()
    'maintain the width:height ratio when the form is resized using a mouse
    StretchShrink Me
End Sub
```

```
Sub GetOriginalSize(frm As Access.Form)
    '-----
    'Procedure :      GetOriginalSize
    'DateTime :      23/12/2022
    'Author :        Colin Riddington
    'Purpose :       Used to get original size of form & its sections before resizing using a mouse
    '-----

    On Error Resume Next

    DoCmd.Restore
```

```
'get form dimensions on load
getOrigWindow frm

'get height of each section
lngHeaderHeight = frm.FormHeader.Height
lngFooterHeight = frm.FormFooter.Height
lngDetailHeight = frm.Detail.Height

'on form load set both lngNew & lngOld values to equal lngOrig values
lngNewWidth = lngOrigWidth
lngOldWidth = lngOrigWidth

lngNewHeight = lngOrigHeight
lngOldHeight = lngOrigHeight

End Sub
```

```
Sub StretchShrink(frm As Access.Form)
    '-----
    'Procedure :      StretchShrink
    'DateTime :      23/12/2022
    'Author :        Colin Riddington
    'Purpose :       Used to maintain width:height ratio when the form is resized using a mouse
    '-----
```

```
On Error Resume Next
```

```
Dim sngStretchWidth As Single, sngStretchHeight As Single
Dim blnRestore As Boolean
```

```
Application.Echo False
```

```
'ensure width & height are changed in proportion as form is stretched / shrunk
GetResizedWindow frm
```

```
sngStretchWidth = lngNewWidth / lngOldWidth
sngStretchHeight = lngNewHeight / lngOldHeight
```

```
If sngStretchWidth >= sngStretchHeight Then
    If sngStretchWidth >= 1 Then
        sngStretchHeight = sngStretchWidth
        lngNewHeight = lngOldHeight * sngStretchHeight
    Else
        sngStretchWidth = sngStretchHeight
        lngNewWidth = lngOldWidth * sngStretchWidth
    End If
```

```
Else
    If sngStretchHeight >= 1 Then
        sngStretchWidth = sngStretchHeight
        lngNewWidth = lngOldWidth * sngStretchWidth
    Else
        sngStretchHeight = sngStretchWidth
        lngNewHeight = lngOldHeight * sngStretchHeight
    End If
```

```
End If
```

```
DoEvents
'move to top left & update form size
DoCmd.MoveSize 0, 0, lngNewWidth, lngNewHeight
```

```
'adjust height of each section in proportion
If GetStretchFactor <> 1 Then 'resize necessary
    frm.Detail.Height = lngDetailHeight * GetStretchFactor
    frm.FormHeader.Height = lngHeaderHeight * GetStretchFactor
    frm.FormFooter.Height = lngFooterHeight * GetStretchFactor
```

```
'resize all controls
Resize GetStretchFactor, frm
```

```
End If
```

```
'set old values to equal new values (used in GetStretchFactor)
lngOldWidth = lngNewWidth
lngOldHeight = lngNewHeight
Application.Echo True
```

```
End Sub
```

Automatic Form Resizing in Access

Datasheet (sub)form code:

```
Private Sub Form_Load()

'ensure datasheet opens with correct size font and correct column widths
Me.DatasheetFontHeight = 10 'deliberately small for this DEMO

'resizing handled in main form if used as a subform
'If standalone form, adjust font size using GetCurrentFactor
If Not IsSubform(Me) Then
    intFontSize = Me.DatasheetFontHeight
    Me.DatasheetFontHeight = CInt(intFontSize * GetCurrentFactor)
End If

'now loop through controls and adjust column widths automatically
For Each ctrl In Me.Controls
    Select Case ctrl.ControlType
        Case acTextBox, acCheckBox, acComboBox 'limit to prevent error 438 with unsupported control types
            ctrl.ColumnWidth = -2 'autosize width
    End Select
Next

End Sub
```

1

2

Main form code:

```
Dim intFontSize As Single

Private Sub Form_Load()
    ReSizeForm Me
    cmdClose.SetFocus

    intFontSize = Me.fsubControlTypesDS.Form.DatasheetFontHeight
    'Debug.Print intFontSize
    Me.cmdResizeDS.Caption = "Resize Datasheet"

    'show screen resolution
    Me.lblResolution.Caption = GetResolution() & " : " & GetScreenShape() & " (" & GetFormFactor() & ")"
End Sub
```

3

Main form code (continued):

```
Private Sub cmdResizeDS_Click()

Me.cboZoom = 100
cboZoom_AfterUpdate

If Me.cmdResizeDS.Caption = "Resize Datasheet" Then
    Me.cmdResizeDS.Caption = "UnResize Datasheet"
    Me.fsubControlTypesDS.Form.DatasheetFontHeight = Me.fsubControlTypesDS.Form.DatasheetFontHeight * GetCurrentFactor
Else
    Me.cmdResizeDS.Caption = "Resize Datasheet"
    Me.fsubControlTypesDS.Form.DatasheetFontHeight = Me.fsubControlTypesDS.Form.DatasheetFontHeight / GetCurrentFactor
End If

For Each ctrl In Me.fsubControlTypesDS.Form.Controls
    'loop through controls and adjust column widths automatically
    Select Case ctrl.ControlType
        Case acTextBox, acCheckBox, acComboBox 'limit to prevent error 438
            ctrl.ColumnWidth = -2 'autosize width
    End Select
Next

End Sub
```

4

5

```
Private Sub cboZoom_AfterUpdate()


If Me.cmdResizeDS.Caption = "Resize Datasheet" Then
    Me.fsubControlTypesDS.Form.DatasheetFontHeight = intFontSize * cboZoom / 100
Else
    Me.fsubControlTypesDS.Form.DatasheetFontHeight = intFontSize * GetCurrentFactor * cboZoom / 100
End If

For Each ctrl In Me.fsubControlTypesDS.Form.Controls 'set widths automatically
    'loop through controls and adjust column widths automatically
    Select Case ctrl.ControlType
        Case acTextBox, acCheckBox, acComboBox 'limit to prevent error 438 with unsupported control types
            ctrl.ColumnWidth = -2 'autosize width
    End Select
Next

End Sub
```


Automatic Form Resizing in Access


AFR tutorial: <https://isladogs.co.uk/automatic-form-resizing-1/>




Isladogs on Access

Free examples, samples and guidance for developers

[Home](#) [Applications](#) [Purchase](#) [Downloads](#) [Resources](#) [Services](#) [Search](#)





ResizeForm Me - A Tutorial in Automatic Form Resizing Page 1 (of 4)

[Page 2](#) [Page 3](#) [Page 4](#)

Version 3.93 First Published 10 Mar 2019 Last Updated 3 Mar 2023 Difficulty level: **Moderate**

Section Links (this page):

- [Introduction](#)
- [What is AFR?](#)
- [Layout Guides](#)
- [Using Automatic Form Resizing](#)
- [Design with the form resizer](#)
- [Set default form size](#)
- [Downloads](#)
- [YouTube Videos](#)
- [Summary/Review](#)
- [Feedback](#)

1. Introduction [Return To Top](#)

The previous update to **version 3.90** provided a **major update** to my series of articles about **automatic form resizing** in Access

In that version, the **example database** was **significantly updated** with **many new features** including:

- a) support for **high resolution monitors** e.g. 3840 x 2160 to handle **possible issues** related to the **maximum permitted form width**
- b) improved support for **portrait monitors** e.g. 1080 x 1920 resolution
- c) added code to allow forms & controls to be **stretched/shrunk 'on the fly'**
- d) added code to **refresh display** when forms are moved to a **secondary monitor** with a **different size/resolution**
- e) improvements to **zoom form code**
- f) added code to allow scaling of **command button images**
- g) two additional template forms

All three pages of the **existing article** were updated with **additional information** and **new images**.
In addition, a completely **new page** was added which discusses the **latest features** in detail.

Automatic Form Resizing in Access

New web page for today's session: <https://isladogs.co.uk/using-afr>

Isladogs on Access

Free examples, samples and guidance for developers

Home Applications Purchase Downloads Resources Services Search

Making Automatic Form Resizing Work For You

First Published 28 April 2023

I was very pleased to be invited to give a presentation on 28 April 2023 to the **Access DevCon** online conference organised by **Karl Donaubauer (MVP)**. **Access DevCon** is an annual two day event and is by far the largest English language Access conference. This year there were around 160 participants from 22 countries spanning 5 continents.

My presentation was called [Making Automatic Form Resizing Work For You](#)

Topic Outline

My free **automatic form resizing** (AFR) code has been in continuous development for well over 15 years. For example, in recent years, I have added support for **tabbed documents display together with resizing datasheets, split forms** and **navigation forms**.

The first part of this session was used to demonstrate some of the **latest features** available in my AFR code including:

- Support for **high resolution monitors**
- Improved support for **portrait monitors**
- **Stretch/shrink forms** 'on the fly'
- **Refresh display** after change of **resolution / monitor**
- **Image scaling** on **command buttons**

Following that, in response to a challenge set by Karl, I demonstrated live how easy it is to use AFR to successfully convert **existing applications** by focusing on 3 applications:

- An 'oversized form' originally created in a high resolution
- My Better Date Picker utility which includes 42 date buttons, each redrawn after every change of month/year
- An old Northwind database created for Access 2000 in 1024*768 resolution

The intention was that after the session, anyone would be able to integrate my code into their own applications and have them automatically resized for all monitors and resolutions.

Below are some links to files used during the presentation which you can download